A. Veneziani -  Considerazioni sullo svolgimento esercizio “distributore    automatico”


Considerazioni generali

Un distributore automatico di bevande è assimilabile ad un automa a stati finiti (nella categoria degli automi a stati finiti possono essere assimilate anche altre macchine di uso comune tra cui ad, esempio, le lavatrici, le lavastoviglie e gli stessi calcolatori elettronici) e perciò anche un sistema automatico.

Cogliamo quindi l’occasione di simulare tale sistema con il calcolatore, abbinando, quando possibile, le due materie di Informatica e Sistemi automatici.

La simulazione di un distributore può raggiungere ovviamente gradi più o meno alti di similitudine alla realtà, ed ovviamente spesso una simulazione più esatta comporta codice maggiore.

La versione qui presentata può essere considerata di media complessità; è possibile immaginare e programmare versioni più difficili, ma anche più schematiche e facili.

Il programma fa un uso intensivo di option button, sia per la scelta della bevanda, che per l’indicazione della moneta inserita in un certo momento.  Opportuno quindi comprendere a fondo questo elemento delle GUI, che ha una specifica programmazione.

Il diagramma degli stati di un distributore

Per chiarire il problema e migliorare l’analisi dello stesso può essere utile, essendo il distributore un macchina a stati finiti, scrivere il diagramma degli stati di questa macchina.  Individuando gli stati possibili, schematizziamo il problema e ciò ci aiuta nella soluzione rigorosa dello stesso.

Indichiamo quindi il diagramma:

Indichiamo sinteticamente il significato delle singole freccie che identificano i cambiamenti di stato:

1. Passaggio dello stato inattivo (spento) a quello di pronto all’uso

2. Procedimento di conteggio dell’importo totale, attivato dall’inserimento di una moneta

3. L’utente passa a selezionare una bevanda.  Questo cambiamento di stato non è tanto della macchina, quanto del comportamento dell’utilizzatore (smette di inserire monete e attiva una selezione)

4. Transizione che avviene solo in presenza di importo insufficiente alla luce del costo della bevanda selezionata

5. Il distributore recepita la selezione e ritenuto l’importo adeguato o in eccesso, passa ad effettuare l’erogazione

6. L’erogazione dopo un certo tempo ha termine e la macchina ripassa ad aspettare un nuovo inserimento di monete

Gli option (radio) button

Ripetiamo che la selezione del tipo di moneta, avviene tramite selezione con radio (o option che dir si voglia) button;  questi, usualmente, su una form sono suddivisi in gruppi, che possono operare in modo indipendente da gruppo a gruppo.

All’interno di ogni gruppo, invece, solo uno degli option button può risultare selezionato, mentre gli altri risulteranno necessariamente tutti deselezionati (per questo assumono anche il nome di radio button, in quanto le vecchie radio avevano tastiere di comando che funzionavano spesso con la stessa logica, ogni tasto premuto deselezionava gli altri).  Data questa regola non ha ovviamente senso avere un option button isolato, ma devono essere raggruppati almeno due elementi.

Ogni option button ha una label abbinata, in cui poter scrivere la descrizione della selezione.  Il contenuto è regolato dalla proprietà Caption.

Ogni gruppo di option button è definito inserendo gli option button  in un apposito controllo-contenitore detto Frame.  Ossia gli option button (e solo quelli) presenti nello stesso frame sono interdipendenti.  Questo indipendentemente dal nome ad essi assegnato. Il frame rappresenta un elemento indipendente dove possono essere “agganciati” i controlli. Esso è usualmente delimitato da un bordo visibile. Ha un titolo regolabile con la proprietà Caption del frame stesso.

Usualmente ad uno stesso gruppo di option button viene assegnato lo stesso nome, ed essi vengono dichiarati quindi come array di controlli.  In questo caso il nome dei vari option button è il medesimo (ad esempio Moneta).

Ma come distinguere tra i vari componenti del gruppo ?   Questo è necessario  per capire ad esempio quale di essi è selezionato.   Per risolvere il problema esiste un indice (indicato dalla proprietà Index) numerico intero positivo >= 0 che enumera tutti i componenti dell’array. Questi componenti, proprio come in una matrice (array in inglese) sono individuati tramite un indice numerico.

In pratica il gruppo di option button Moneta avrà per ogni option lo stesso nome, ma per ogni elemento varierà l’indice Index, che assumerà per i vari componenti 0, 1, 2… a seconda di quanti elementi ci sono.

Gli option button sono quasi sempre raggruppati in array di elementi; più raro che siano usati array per altri tipi di elementi grafici (bottoni ecc.).

E’ comunque possibile e utile, in alcune situazioni, avere array anche di questi elementi.  In questo caso si segue la procedura appena detta sopra per definire l’array, mantenendo lo stesso nome e variando il valore dell’indice.   

Timer in VB

Un altro elemento nuovo del progettino della macchina distributrice è il controllo Timer.  Come detto in seguito, esso è indirettamente un modo per implementare, in qualche modo, anche capacità di elaborazione parallela in Visual Basic.   Il controllo Timer và attivato tramite una apposita proprietà Enabled che và messa a true (attivo) o false (disattivo) a seconda dei casi.  Associato al timer c’e’ un apposito evento che viene attivato ad intervalli regolari in base all’indicazione in  millisecondi inserita nella proprietà Interval, che come dice il suo nome indica l’intervallo tra due chiamate a tale evento.  Nel nostro caso non ci serve un intervento ripetuto ad intervalli regolari del timer, ma solo un ritardo opportunamente effettuato al momento in cui si comanda la preparazione della bevanda.  Ciò può essere svolto con un timer, che al momento del primo evento viene spento.  Il programma poi si occupa di una nuova accensione (inizio preparazione bevanda) in modo opportuno.

Come si può capire anche il controllo timer può essere utile in svariate situazioni e problemi.

Il controllo Timer è visibile solo a “design time”, ma non è visibile a “run time”.

L’inserimento delle monete

Premesso quanto sopra, innanzittutto si nota, al momento della pressione del tasto che simula l’inserimento di una moneta, che esiste un ciclo di scansione dei radio button per determinare quale risulti selezionato.

In effetti infatti tutti gli option button risulteranno avere valore non selezionato e solo quello scelto avrà valore selezionato.  La domanda che viene spontanea è come fare ad individuare quello selezionato.  La via più veloce per saperlo è utilizzare lo stesso sistema ad eventi (specificamente l’evento click) che è associato al gruppo degli option button.  A questo evento infatti è associato un indice che informa sulla selezione appena effettuata.  Pur tuttavia qui si è utilizzato un metodo che non chiama in causa l’evento click del gruppo di option.  Praticamente ad un certo momento si scandisce tutti gli option per vedere quale di loro risulta selezionato. 

Come sia possibile fare questo controllo è piuttosto evidente dal listato: la proprietà testata è la Value ed essa assume valori booleani (True – selezionato  o False -  non selezionato).

In tal modo una apposita variabile permette di comprendere quale option sia selezionato.   Successivamente si individua il valore in euro della moneta rappresentata dall’option button selezionato.  Questa selezione sfrutta una  istruzione apposita detta Select Case.  Con altri nomi questa stessa istruzione è presente anche in C, Pascal standard, Java e altri linguaggi più o meno moderni.

Si prega l’alunno di rilevare la sintassi dell’istruzione stessa sul manuale di Visual Basic una volta in laboratorio per considerare:

1. I tipi di dato che possono essere oggetto della selezione .

2. Le condizioni inseribili, che hanno regole diverse da quelle valide per la If.  

3. Il modo di concatenare condizioni multiple

4. Il modo di definire un evenienza di default (Else Case), che venga eseguita ogni volta che tutte le altre sono risultate non verificate

La selezione avviene rilevando il valore di una variabile Select …. E compiendo operazioni di volta in volta diverse a seconda dei valori Case…

Si tratta in definitiva di una cosiddetta selezione multipla, ossia una parte di programma dove più possibili evenienze possono verificarsi.  Se l’istruzione If è spesso paragonata ad un bivio, qui ci troviamo di fronte ad un incrocio da dove si diramano più strade.

Dal listato la routine per la selezione e l’inserimento di monete risulta:

' Evento equivalente all'inserzione di monente...

Private Sub cmd_Inserisci_Click()

Dim Valore_Moneta As Currency

Dim Selezionata As Integer

Dim Scan As Integer

    ' individua il tipo di moneta inserito...

    For Scan = 0 To 5

        If Monete(Scan) = True Then Selezionata = Scan

    Next

    ' associa alla moneta inserita il suo valore in euro...

    Select Case Selezionata

       Case 0:  Valore_Moneta = 0.05

       Case 1:  Valore_Moneta = 0.1

       Case 2:  Valore_Moneta = 0.2

       Case 3:  Valore_Moneta = 0.5

       Case 4:  Valore_Moneta = 1

       Case 5:  Valore_Moneta = 2

    End Select

    ' cumula in un totale l'importo delle monete inserite

    Importo = Importo + Valore_Moneta

    txt_Totale.Text = Importo

End Sub

In questo caso questa selezione risulta assegnare a Valore_moneta un valore opportuno (pari a quello della moneta inserita – anche questa  è una simulazione, in quanto non si tratta della moneta effettiva, ma di una sua astrazione rappresentata dal suo valore).  Una volta determinato questo si cumula il valore in una variabile atta a contenere l’importo totale.  

Tale importo viene immediatamente anche visualizzato nella apposita box di visualizzazione (si consiglia, per le box in cui si và solo a scrivere, come questa, di utilizzare label, oppure di fissare la proprietà Locked a True.  Ciò evita che il contenuto possa essere alterato.   

Successivamente fissato tramite più inserimenti di monete un importo, si passa a selezionare la bevanda voluta:

' evento associato alla scelta di una bevanda

Private Sub Bevanda_Click(Index As Integer)

Dim OkPrep As Boolean

Dim Costo As Currency

Dim Scan As Integer

    OkPrep = True

    ' definisce il costo della bevanda in base alla selezione

    Select Case Index

        Case 0:  Costo = 0.3

        Case 1:  Costo = 0.35

        Case 2:  Costo = 0.3

        Case 3:  Costo = 0.4

    End Select

    ' controlla il caso di importo insufficiente...

    If Importo < Costo Then OkPrep = False

    ' se l'importo e' sufficiente (OkPrep True)...

    If OkPrep And (cmd_PrelievoBev.Enabled = False) Then

        txt_Totale.Text = ""

         Timer1.Enabled = True

        lbl_Out.Caption = "Bevanda in preparazione..."

        ' calcola l'entità del resto e lo visualizza...

         Resto = Importo - Costo

         If txt_Resto.Text <> "" Then

              txt_Resto.Text = CCur(txt_Resto.Text) + Resto

         Else

              txt_Resto.Text = Resto

         End If

         Importo = 0

    Else

        If (cmd_PrelievoBev.Enabled = True) Then

            lbl_Out.Caption = "--> Ritirare bevanda. <--"

            Beep

        Else

            ' avverte che l'importo non e' sufficiente

            lbl_Out.Caption = "Importo insufficiente."

            Beep

            For Scan = 0 To 3

                Bevanda(Scan) = False

            Next

        End If

    End If

End Sub

Questa routine più complessa della precedente è associata invece direttamente  agli option button del selettore del tipo di bevanda.  Questo gruppo di selettori è ovviamente indipendente da quello delle monete.  L’individuazione dell’option selezionato in questo caso è automatica e avviene tramite l’evento stesso.  Il click su uno dei pulsanti radio, scatena l’evento, ma nel contempo il parametro Index dell’evento sta ad indicare quale option button sia stato selezionato.

In base a questo valore (intero) viene effettuata un’altra selezione multipla che consente di fissare il prezzo della bevanda.  Si controlla subito dopo, avendo definito il prezzo ed avendo calcolato l’importo precedentemente, se la bevanda sia o no acquistabile.  Questo viene indicato dalla variabile OkPrep.  Se il prezzo è ammissibile (ossia abbiamo dato sufficienti monete), e la bevanda preparata in precedenza è stata prelevata, allora si può passare alla erogazione di una nuova bevanda. In questo caso si attiva un Timer che produrrà gli effetti della bevanda pronta con un certo ritardo.  Questo timer in realtà implementa un processo di conteggio del tempo parallelo ed indipendente dal programma principale.  Si tratta dell’unica forma nativa di “multithreading” di Visual Basic.  Terminato questo tempo viene attivato il pulsante preposto al prelievo (simula il prelievo a mano), e attiva la scritta che la bevanda è pronta.

Finchè ciò non accade però, subito dopo l’avvio della produzione di una certa bevanda, appare la scritta “Bevanda in preparazione….”  e viene effettuato ed erogato il resto.   Queste sono operazioni svolte dal programma principale nel codice di questo evento che si sta considerando (selezione della bevanda). 

lbl_Out.Caption = "Bevanda in preparazione..."

' calcola l'entità del resto e lo visualizza...

Resto = Importo - Costo

Una apposita If si cura di cumulare il resto, se è già  presente una cifra, o altrimenti di registrarlo se il resto iniziale risultava 0 (condizione normale).

If txt_Resto.Text <> "" Then

     txt_Resto.Text = CCur(txt_Resto.Text) + Resto

Else

     txt_Resto.Text = Resto

End If

Importo = 0

Infine dopo tale operazione viene azzerato l’importo i quanto ormai utilizzato per produrre bevanda e resto.

Nel caso invece la bevanda non sia erogabile:

If OkPrep And (cmd_PrelievoBev.Enabled = False) Then

...

Else

        If (cmd_PrelievoBev.Enabled = True) Then

            lbl_Out.Caption = "--> Ritirare bevanda. <--"

            Beep

        Else

            ' avverte che l'importo non e' sufficiente

            lbl_Out.Caption = "Importo insufficiente."

            Beep

            For Scan = 0 To 3

                Bevanda(Scan) = False

            Next

        End If

    End If

End Sub

Il programma di controllo si chiede se la bevanda sia effettivamente stata prelavata.  Se non lo è stata, ricorda all’utente di farlo, ma non procede con una nuova preparazione.  Ossia il comando di preparazione non ha effetto.   Altrimenti si può solo concludere che l’importo era insufficiente e questo viene scritto nel messaggio della apposita label.  In entrambi i casi viene prodotto un beep (istruzione Beep del Visual Basic), che evidenzia la situazione di errore.   Infine il selettore delle bevande in entrambi i casi viene resettato.  Ciò permette di riavere un nuovo evento selezione successivamente.  Lo svantaggio è che non resta traccia (come nella realtà del resto, mi risulta) della selezione effettuata.

Infine vi sono le parti che simulano il prelievo del resto e quello della bevanda simulati con due pulsanti.  Al momento che queste routine entrano in azione vengono effettuate alcune operazioni.

Per il prelievo bevanda:  

· Il pulsante viene disattivato (non si può prendere la bevanda 2 volte !)

· Cambia al messaggio “inserire i soldi…” il messaggio del distributore

· Vengono deselezionati tutti i selettori

' tasto che simula il prelevamento dell bevanda...

Private Sub cmd_PrelievoBev_Click()

Dim Scan As Integer

    ' dopo il prelevamento viene visualizzato un

    ' messaggio che ricorda di inserire il denaro...

    lbl_Out.Caption = "Inserire i soldi..."

    ' reset di tutte le selezioni...

    For Scan = 0 To 3

        Bevanda(Scan) = False

    Next

    For Scan = 0 To 5

        Monete(Scan) = False

    Next

    ' il pulsante che comanda il prelievo deve risultare

    ' disattivato (non c'e' piu' nulla da prelevare)

    cmd_PrelievoBev.Enabled = False

End Sub

Per il prelievo del resto:

· Si azzera la variabile che conteggia il resto

· Si aggiorna l’indicazione del resto nella label o textbox che sia preposta a ciò

' simula il prelievo del resto...

Private Sub cmd_PrelievoResto_Click()

    Resto = 0

    txt_Resto.Text = ""

End Sub

Il timer  di ritardo invece una volta raggiunto il tempo di preparazione da luogo a:

· Indicare il promemoria nella label o textbox di ritirare la bevanda

· Attiva il pulsante di prelievo della bevanda (una bevanda è stata appena preparata e quindi è prelevabile)

· Spegne l’attività del timer (non dovrà per il momento effettuare ulteriori conteggi di tempo, fino a nuovo ordine, ossia quando Enabled = true)

' simula la preparazione della bevanda introducendo

' un opportuno tempo di ritardo tra la selezione

' e la possibilità di ritiro della bevanda stessa

Private Sub Timer1_Timer()

    lbl_Out.Caption = "Ritirare bevanda e resto."

    Timer1.Enabled = False

    cmd_PrelievoBev.Enabled = True

End Sub
Inserimento


monete





Accensione





Selezione bevanda





Erogazione bevanda





1





2





3





4





5





6








Revisione 2                                                                          3                                                       Analisi distributore automatico

